RESOLUTIONS OVER POLYNOMIAL RINGS

Questions

(1) Pick your favourite prime number p and consider the ring

$$R = \frac{\mathbb{F}_p[x,y]}{\big(xy(x^{p-1} - y^{p-1})\big)}\,.$$

   (a) How do you use [?M2] to check that your favourite number is prime?
   (b) Show that variety $\mathbb{V}\big(xy(x^{p-1}-y^{p-1})\big)$ contains all the $\mathbb{F}_p$-rational points lying on the projective line over $\mathbb{F}_p$. Therefore, there cannot be a linear nonzerodivisor.
   (c) What are the dimension and depth of R?
   (d) Find a homogeneous nonzerodivisor in R.

   Hint. Use the `random` function.

(2) (a) In [?M2], construct the Koszul complex for the monomial basis for $\big(\mathbb{Q}[x,y]\big)_5$.

   Hint. One method involves constructing a homogeneous map between polynomial rings.

   (b) Study the homology of this complex (e.g. vanishing, Hilbert Series, etc.)

(3) Let $S = \mathbb{Q}[x_0, x_1, x_2, x_3]$.
   (a) Let M be the image of the middle differential in Koszul complex on the variables. Determine the endomorphism ring E of M over S. As an S-module, what are the rank, depth, betti numbers and Hilbert series of E?
   (b) Determine the homology of the dual of the resolution of E. What are the dimension and Hilbert series of the homology modules? Explain why E is locally free outside the ideal $(x_0, \ldots, x_3)$. Why is M locally free outside the ideal $(x_0, \ldots, x_3)$?

(4) Let I_n denote the ideal of $(n \times n)$ commuting matrices.
   (a) What is the "expected" dimension of S/I_n?
   (b) Fix $n = 3$ and let J be the "off-diagonal" ideal. Compute $I' := J : I_3$ and show that S/I' is Cohen-Macaulay.
   (c) (Open?) How many components does the variety $\mathbb{V}(I')$ have? In other words, how many minimal primes lie over I'?
   (d) Find 12 (random?) linear forms that form a regular sequence on S/I_3.

Date: 25 July 2006.

(5) Let p be a prime number and consider the following polynomials in $\mathbb{F}_p[x]$:

$$f = x^8 + x^6 + 10x^4 + 10x^3 + 8x^2 + 2x + 8$$
$$g = 3x^6 + 5x^4 + 9x^2 + 4x + 8$$

(a) Compute the continued fraction expansion for g/f.

Hint. In [?M2], `f // g` gives the quotient and `f % g` gives the remainder.

(b) Homogenize f and g to obtain f^h and $g^h \in \mathbb{F}_p[x, y]$ and set $I_j := (f^h, g^h, y^j)$ for $1 \leq j \leq 13$. Compute the minimal free resolution of each these ideals — in particular, examine the maps.

(c) Repeat part (b) with $p = 13$.

(d) Explain the relationship between the Hilbert-Burch matrix and the continued fraction expansion.

Macaulay 2 examples from the morning lecture

```
-- resolutions for powers of maximal ideal
S = QQ[x,y];
powerIdeal = d -> res ((ideal gens S)^d);
scan(1..2, i -> (
  C1 := powerIdeal (3*i-2);
  C2 := powerIdeal (3*i-1);
  C3 := powerIdeal (3*i);
  << endl << betti C1 << "      "
  << betti C2 << "     "
  << betti C3 << endl))
scan(1..2, i -> (
  C1 := powerIdeal (3*i-2);
  C2 := powerIdeal (3*i-1);
  C3 := powerIdeal (3*i);
  << endl << C1.dd_2 << "      "
  << C2.dd_2 << "     "
  << C3.dd_2 << endl))

-- resolution of twisted cubic
S = QQ[w,x,y,z];
M = matrix{{w,x,y},{x,y,z}}
twistedCubic = minors(2,M)
twistedCubic == monomialCurveIdeal(S,{1,2,3})
F = res (S^1/twistedCubic)
betti F
F.dd
```

```
-- find nonzero divisors
prune Tor_1(S^1/twistedCubic, S^1/ideal(w))
prune Tor_1(S^1/(twistedCubic + ideal(w)), S^1/ideal(z))
-- relating twisted cubic to square of maximal ideal
mingens(twistedCubic + ideal(w,z))

-- ideal of commuting 2*2 matrices
S = ZZ/101[a_1..a_4,b_1..b_4];
A = genericMatrix(S,2,2)
B = genericMatrix(S,b_1,2,2)
com2 = ideal flatten entries (A*B-B*A)
F = res (S^1/com2)
betti F
mingens com2

-- ideal of commuting 3*3 matrices
S = ZZ/101[a_1..a_9,b_1..b_9];
A = genericMatrix(S,3,3)
B = genericMatrix(S,b_1,3,3)
com3 = ideal flatten entries (A*B-B*A)
F = res (S^1/com3)
betti F
codim (S^1/com3)
dim (S^1/com3)

-- ideal of "off diagonal entries" in commuting 3*3 matrices
offDiag = ideal flatten apply(3,
     i -> apply(toList(0..i-1|i+1..2),
  j -> (A*B-B*A)_(i,j)));
betti res offDiag

-- invariants of twisted cubic
S = ring twistedCubic;
hilbertSeries (S^1/twistedCubic)
reduceHilbert hilbertSeries (S^1/twistedCubic)
hilbertPolynomial(S^1/twistedCubic)
hilbertPolynomial(S^1/twistedCubic, Projective => false)

-- invariants of minimal surface
S = QQ[a_1..a_6];
A = genericSymmetricMatrix(S,3)
symMin = minors(2,A)
```

```
betti res symMin
reduceHilbert hilbertSeries (S^1/symMin)
hilbertPolynomial(S^1/symMin, Projective => false)

-- invariants of maximal minors
R = QQ[b_1..b_8];
B = genericMatrix(R,2,4)
genMin = minors(2,B)
betti res genMin
reduceHilbert hilbertSeries (R^1/genMin)
hilbertPolynomial(R^1/genMin, Projective => false)

-- invariants of commuting 2*2 matrices
S = ring com2;
reduceHilbert hilbertSeries (S^1/com2)
hilbertPolynomial(S^1/com2, Projective => false)

-- invariants of commuting 3*3 matrices
S = ring com3;
reduceHilbert hilbertSeries (S^1/com3)
hilbertPolynomial(S^1/com3, Projective => false)

-- Koszul complex
S = QQ[a_1..a_6];
koszul(3, matrix{gens S})
-- compare with differential in resolution of offDiag
S = ring offDiag;
(res offDiag).dd_3

-- betti numbers of twistedCubic via Koszul complex
S = ring twistedCubic;
K = res ideal gens S
C = K ** (S^1/twistedCubic);
prune HH(C)
apply(1+length C,
     i -> reduceHilbert hilbertSeries HH_i(C))

-- check if twistedCubic is Cohen-Macaulay
F = res (S^1/twistedCubic)
G = Hom(F,S^1)
prune HH(G)
```

```
-- something that is not Cohen-Macaulay
quartic = monomialCurveIdeal(S,{1,3,4})
hilbertPolynomial(S^1/quartic, Projective => false)
F = res(S^1/quartic)
G = Hom(F,S^1)
prune HH(G)

-- check if "com2" is Cohen-Macaulay
S = ring com2;
F = res (S^1/com2)
G = Hom(F,S^1)
prune HH(G)
```